

TRAVAUX PRATIQUES N° 1. — PROCESSUS DE LÉVY RÉELS

1. Introduction théorique

Un processus réel $X = (X_t)_{t \geq 0}$ est un processus à accroissements indépendants si et seulement si pour tous $0 = t_0 < t_1 < \dots < t_n$, $X_0, X_{t_1} - X_0, \dots, X_{t_n} - X_{t_{n-1}}$ sont des variables aléatoires réelles indépendantes.

Les accroissements sont homogènes si et seulement si la loi de $X_t - X_s$ ne dépend que de $t - s$, on la note alors μ_{t-s} . Les lois $(\mu_t)_{t \geq 0}$ forment alors un semi-groupe de convolution : $\mu_0 = \delta_{\{0\}}$ et $\mu_s * \mu_t = \mu_{s+t}$ pour tous $s, t \geq 0$ (de manière équivalente leurs fonctions caractéristiques vérifient $\varphi_s \times \varphi_t = \varphi_{s+t}$ pour tous $s, t \geq 0$).

Lorsque le semi-groupe $(\mu_t)_{t \geq 0}$ satisfait de plus la condition de continuité $\mu_t \rightarrow \mu_0$ quand t tend vers 0 — ce qui équivaut à la continuité à droite du semi-groupe —, le processus X est alors qualifié de processus de Lévy. Cette condition de continuité se lit aussi sur les fonctions caractéristiques $(\varphi_t)_{t > 0}$ puisqu'il faut et suffit que celles-ci convergent simplement vers la fonction constante égale à 1 (fonction caractéristique de $\delta_{\{0\}}$) lorsque $t \rightarrow 0$ d'après le théorème de Lévy.

Les processus à accroissements indépendants homogènes satisfont deux types d'homogénéité : l'homogénéité temporelle (invariance par translation dans le temps : ce sont des processus de Markov dans leur filtration naturelle), et l'homogénéité spatiale (invariance par translation dans l'espace). Ces deux propriétés les caractérisent. Le semi-groupe de transition $(P_t)_{t \geq 0}$ est alors lié au semi-groupe de convolution par

$$P_t(x, B) = \mathbb{P}^x\{X_t \in B\} = \mathbb{P}^x\{X_t - x \in B - x\} = \mu_t(B - x).$$

La composition des noyaux

$$\begin{aligned} (P_s \circ P_t)(x, B) &= \int_{\mathbb{R}} P_s(x, dy) P_t(y, B) = \int_{\mathbb{R}} \mu_s(dy - x) \mu_t(B - y) \\ &= \int_{\mathbb{R}} \mu_s(dy) \mu_t((B - x) - y) = (\mu_s * \mu_t)(B - x) = P_{s+t}(x, B), \quad x \in \mathbb{R}, B \in \mathcal{B}(\mathbb{R}), \end{aligned}$$

correspond bien à la convolution des mesures de probabilité dans ce cas.

Les semi-groupes de convolution de mesures de probabilité sur \mathbb{R} (et \mathbb{R}^n) continus en 0, c'est-à-dire continus à droite, ont été caractérisés par Paul Lévy, en particulier le semi-groupe multiplicatif $(\varphi_t)_{t \geq 0}$ de leurs fonctions caractéristiques vérifient $\varphi_t(\theta) = \mathbb{E}^0[e^{i\theta X_t}] = e^{tg(\theta)}$ où

$$g(\theta) = ia\theta - \frac{1}{2}\sigma^2\theta^2 + \int_{|x| \geq m} (e^{i\theta x} - 1) \nu(dx) + \int_{|x| < m} (e^{i\theta x} - 1 - i\theta x) \nu(dx)$$

qui est la formule de Lévy–Khinchine (a est le coefficient de dérive [*drift*], σ le coefficient de diffusion, et ν la mesure de Lévy associés au processus). Parmi eux, nous avons (entre autres) :

- le semi-groupe brownien, $\mu_t(dx) = e^{-x^2/2t} \frac{dx}{\sqrt{2\pi t}}$, $t \geq 0$;
- le semi-groupe de Poisson d'intensité $\lambda > 0$, $\mu_t(dx) = \sum_{n=0}^{\infty} e^{-\lambda t} \frac{(\lambda t)^n}{n!} \delta_{\{n\}}(dx)$, $t \geq 0$.

Si nous savons que le mouvement brownien admet des versions à trajectoires continues, un processus de Poisson est essentiellement un processus de comptage et ses trajectoires sont en escaliers avec sauts d'amplitude 1. La condition de continuité à droite ne garantit que l'existence de versions à trajectoires continues à droite et limitées à gauche (càdlàg).

⊗ À quelle condition un processus déterministe est-il un processus de Lévy ? Et si on ôte la condition de continuité à droite ?

⊗ Constater que les combinaisons linéaires de processus de Lévy indépendants sont des processus de Lévy (on se dispensera de vérifier la condition de continuité, qui se traduit au niveau des processus par la continuité à droite en probabilité). Retrouver cette propriété au niveau des semi-groupes de convolution.

⊗ Se souvenir de son cours de L3 ou de M1 pour citer d'autres semi-groupes de convolution classiques (deux au moins) sans avoir lu la suite de ce document.

⊗ Pour comprendre le paramètre de dérive, se souvenir que si une variable aléatoire réelle X_t est intégrable alors sa fonction caractéristique est dérivable en $\theta = 0$ et $\varphi'_{X_t}(0) = i\mathbb{E}[X_t]$. Faire formellement le lien avec la formule de Lévy–Khinchine pour $m = \infty$. (En toute rigueur le coefficient a dépend du choix de m dont le rôle est d'assurer que les intégrales aient un sens.)

2. Simulation de processus à accroissements indépendants homogènes

Fixons un horizon temporel $T = 1$ par exemple. Nous souhaitons représenter une trajectoire d'un processus de Lévy X (ou plus généralement, d'un processus à accroissements indépendants) sur $[0, T]$. Pour ce faire, nous divisons $[0, T]$ en n intervalles de longueur $\Delta t = T/n$, et tirons successivement les n accroissements correspondants. La trajectoire s'obtient alors en cumulant ces accroissements (on peut supposer $X_0 = 0$ puisqu'il y a homogénéité spatiale). Voici un exemple de code pour commencer :

```
// Simulation de processus \a accroissements ind\ependants homog\enes
clear; mode(0);
T = 1; n = 100;
dX = grand(n, 1, "nor", 0, sqrt(T/n));
time = (T/n)*[0:n]'; // vecteur colonne [0; T/n; ...; T*(n-1)/n; T]
X = cumsum([0; dX]); // trajectoire issue de 0
plot2d(time, X);
```

2.1. MOUVEMENT BROWNIEN

⊗ Simuler une trajectoire brownienne sur $[0, T]$ issue de 0 et la représenter par une animation au cours du temps, par exemple de la façon suivante (bricolage maison)

```
// Mouvement brownien
Xmax = ceil(max(X)); Xmin = floor(min(X)); // bornes verticales du graphique
clf();
a = gca(); // get current axes
a.data_bounds = [0, Xmin; T, Xmax]; // clair ?
drawaxis(x=[0:10]*T/10, y=0, dir="d"); // axe horizontal
drawaxis(x=0, y=[0:10]*(Xmax-Xmin)/10+Xmin, dir="l"); // axe vertical
xlabel("Trajectoire brownienne animee");
```

```
for i = 2:n+1; xpause(1);// argument en micro-secondes
    xpoly([time(i-1), time(i)], [X(i-1), X(i)], "lines", 0);
end
```

L'affirmation de l'existence de version continue vous paraît-elle raisonnable ?

⊗ Soit $N = 100$ par exemple. Par une boucle, simuler comme précédemment N trajectoires en ne conservant que les valeurs N terminales. Tracer un histogramme des N valeurs terminales obtenues.

⊗ Comparer la loi de X_T avec les valeurs terminales obtenues via le test de Kolmogorov-Smirnov. On pourra utiliser les programmes `goodness-of-fit.sci` et `kolmogorov.sci` déjà utilisés l'an passé et téléchargeables à l'adresse

<http://www-math.univ-poitiers.fr/~phan/downloads/goodies>

(voir travaux pratiques de 2m12) et la commande `cgofest(<x>, <F>)` où $\langle x \rangle$ est l'échantillon et $\langle F \rangle$ la fonction de répartition de la loi de référence.

⊗ Introduire des paramètres de dérive a et de diffusion σ (on se souvient que la loi $\mathcal{N}(m, \sigma^2)$ a pour fonction caractéristique $\phi(\theta) = \exp(im\theta - \sigma^2\theta^2/2)$). Représenter quelques trajectoires pour voir l'influence de ces paramètres.

2.2. PROCESSUS DE POISSON

La commande `grand(<m>, <n>, <loi>, <paramètres>)` reconnaît la loi de Poisson ("poi") de paramètre λ .

⊗ Représenter une trajectoire du processus de Poisson d'intensité $\lambda = 1$ issu de 0, et stocker N valeurs terminales.

⊗ Que devrait-on noter sur l'allure des trajectoires ?

⊗ Observe-t-on une dérive ? Si oui, que vaut-elle ? En prenant $m = \infty$, identifier tous les termes de la formule de Khinchine dans le cas du processus de Poisson d'intensité λ .

Si on retranche cette dérive au processus, *i.e.* si on considère $(X_t - at)_{t \geq 0}$, quelle est la nature du processus obtenu ?

⊗ Comparer la loi de X_T avec la distribution des valeurs terminales obtenues via le test du χ^2 . La commande `dgofest(<x>, <xtheo>, <ptheo>)` permet de faire la comparaison pourvu qu'on se soit restreint à un nombre convenable de valeurs (aux valeurs théoriques, ou plus exactement classes, doivent correspondre plus de 5 ou 10 valeurs observées).

2.3. PROCESSUS DE CAUCHY

La famille des lois de Cauchy est parfois négligée par les programmeurs. Rappelons que loi de Cauchy de paramètre $a > 0$ est la loi de probabilité absolument continue par rapport à la mesure de Lebesgue admettant la fonction de densité suivante

$$x \in \mathbb{R} \longrightarrow \frac{1}{\pi} \times \frac{a}{a^2 + x^2}.$$

Le calcul de sa fonction caractéristique n'est pas des plus faciles (méthode d'analyse complexe), et donne $\varphi_a(\theta) = \exp(-a|\theta|)$. Il est alors clair qu'en étendant cette famille de mesures de probabilité par $\delta_{\{0\}}$ pour $a = 0$, on obtient un semi-groupe de convolution continu à droite.

⊗ Comblent les lacunes de `grand()` en définissant son propre générateur de lois de Cauchy. Le mettre en œuvre pour réaliser les simulations associées.

⊗ Que peut-on dire des trajectoires ?

⊗ Effectuer le test d'adéquation convenable entre la loi de X_T la distribution des valeurs terminales obtenues.

2.4. PROCESSUS GAMMA

On dispose avec SCILAB de la commande `grand(m, n, "gam", <a>, <lambda>)` pour générer des nombres suivant la loi Gamma de paramètres a et λ . Cette loi est portée par \mathbb{R}_+ et admet pour densité

$$p(x) = \mathbb{1}_{[0, +\infty[}(x) \frac{\lambda^a}{\Gamma(a)} x^{a-1} e^{-\lambda x}, \quad x \in \mathbb{R}.$$

À $\lambda > 0$ fixé, ces lois forment un semi-groupe de convolution indéxé par $a \geq 0$. Ceci peut se voir en calculant la fonction caractéristique d'une loi Gamma de paramètres $\lambda > 0$ et $a > 0$:

$$\varphi(\theta) = \left(\frac{\lambda}{\lambda - i\theta} \right)^a$$

Le cas $a \in \mathbb{N}$ correspondant à la somme de a variables aléatoires indépendantes de loi exponentielle de paramètre λ .

☞ Réaliser les activités précédentes dans le cadre des lois Gamma.

Compléments

LEMME (CRITÈRE DE CONTINUITÉ DE KOLMOGOROV). — Soient $a > 0$, $\gamma > 0$, $\alpha > 1$. Si $(X_t)_{t \in I}$ est une famille de variables aléatoires indexée par un intervalle I de \mathbb{R} telle que :

$$\text{pour tous } t, s \in I, \quad \mathbb{E}[|X_t - X_s|^\gamma] \leq a|t - s|^\alpha, \quad (K_{\gamma, \alpha})$$

alors le processus $(X_t)_{t \in I}$ admet une version (modification) continue

Ce critère n'est qu'une condition suffisante pour qu'un processus admette une version continue. Noter aussi que ce qui compte est ce qu'il se passe lorsque $|t - s|$ est petit.

Dans le cas des processus de Lévy, il suffit de prendre une valeur initiale nulle et de voir si on peut vérifier un critère de la forme

$$\text{pour tout } t \geq 0, \quad \mathbb{E}[|X_t|^\gamma] \leq a|t|^\alpha.$$

Dans le cas brownien, avec $\gamma = 4$, on a $\mathbb{E}[B_t^4] = 3t^2$ (voir travaux dirigés), on peut donc prendre $a = 3$, $\alpha = 2 > 1$. Le cas gaussien plus général peut s'en déduire ($X_t = mt + \sigma B_t$ ou $B_t = (X_t - mt)/\sigma$). Dans le cas des processus de Cauchy de paramètre c , $\mathbb{E}[|X_t|^\gamma] = \infty$ dès que $t > 0$ et $\gamma \geq 1$. Si $0 < \gamma < 1$,

$$\mathbb{E}[|X_t|^\gamma] = 2 \int_0^{+\infty} \frac{(ct)x^\gamma}{(ct)^2 + x^2} \frac{dx}{\pi} = 2 \int_0^{+\infty} \frac{(ct)^{1-\gamma}u^\gamma}{1 + u^2} \frac{du}{\pi} = (ct)^{1-\gamma} \times C(\gamma)$$

qui puisque $0 < 1 - \gamma < 1$ montre que le critère n'est pas non plus vérifiable. Mais ça ne prouve pas que les processus de Cauchy n'aient pas des versions continues... En fait, dans le cas des processus de Lévy, ce critère n'est satisfait que si la mesure de Lévy est nulle (mesure de sauts!), et donc que les lois sont gaussiennes.

RÉFÉRENCES

- [1] BARBÉ (Ph.), LEDOUX (M.), *Probabilité*, EDP Sciences (2007).
- [2] BERTOIN (Jean), *Lévy processes*, Cambridge University Press.
- [3] FOATA (D.), FUCHS (A.), *Processus stochastiques : Processus de Poisson, chaînes de Markov et martingales*, Dunod (2004).
- [4] OUVRARD (J.-Y.), *probabilités 2*, Vuibert (2004).
- [5] CHANCELIER (J.-P.), DELEBECQUE (F.), GOMEZ (C.), GOURSAT (M.), NIKOUKHAH (R.), STEER (S.), *Introduction à Scilab, deuxième édition*, Springer-Verlag France (2007).

TRAVAUX PRATIQUES N° 1. — PROCESSUS DE LÉVY RÉELS, ÉLÉMENTS D'EXPLICATIONS

Ce qui suit tient lieu de trace écrite de ce qui aura été présenté à l'oral plutôt que de compte rendu.

1. Introduction théorique

⊗ Nous avons vu qu'un processus déterministe à valeurs réelles $(x(t))_{t \geq 0}$ est à accroissements indépendants (les constantes étant des variables aléatoires indépendantes de tout ce qu'on veut), et qu'il est homogène si pour tout $0 \leq s \leq t$, $x(t) - x(s) = x(t - s) - x(0)$. En notant $\Delta x(t) = x(t) - x(0)$, $t \geq 0$, on constate que $\Delta x(t)$ est \mathbb{Q}_+ -homogène (si on pose $\Delta x(t) = -\Delta x(-t)$ pour $t \leq 0$, alors Δx est \mathbb{Q} -linéaire). La condition de continuité à droite du semi-groupe de convolution se traduit ici par la continuité à droite de Δx et assure, en passant à la limite, que Δx est \mathbb{R}_+ -homogène : $\Delta x(t) = t \times \Delta x(1)$ pour tout $t \geq 0$, soit $x(t) = x(0) + t \times \Delta x(1)$. Sans hypothèse de régularité aucune, on peut *choisir* des solutions très irrégulières à ce problème, il suffit de considérer \mathbb{R} comme un \mathbb{Q} -espace vectoriel et *choisir* $\Delta x : \mathbb{R} \rightarrow \mathbb{R}$ une application \mathbb{Q} -linéaire différente d'une homothétie.

Évidemment, nous omettons quelques détails sur lesquels le lecteur pourra revenir pour lui-même.

⊗ Il est évident que si $X = (X_t)_{t \geq 0}$ est un processus à accroissements indépendants, homogènes, dont le semi-groupe de convolution vérifie la propriété de continuité à droite en 0, il en est de même de $aX = (aX_t)_{t \geq 0}$ pour toute constante $a \in \mathbb{R}$ (l'écrire proprement).

De même, si X et Y sont deux processus *indépendants* (bien comprendre la force de cette condition), à accroissements indépendants, alors $X + Y$ est un processus à accroissements indépendants, à accroissements homogènes, alors $X + Y$ est à accroissements homogènes, vérifiant la condition de continuité à droite en $t = 0$, alors il en est de même pour $X + Y$ (on pourra noter que la condition de continuité à droite en 0 des semi-groupe équivaut à la continuité à droite en probabilité des processus).

⊗ Quelques semi-groupes de convolution :

- $(\delta_{v \times t})_{t \geq 0}$ le semi-groupe de translation uniforme à vitesse $v \in \mathbb{R}$ (mouvement rectiligne uniforme).
- $(\mathcal{N}(v \times t, \sigma^2 \times t))_{t \geq 0}$ le semi-groupe du mouvement brownien avec dérive (*drift*) de variance $\sigma^2 \geq 0$ (non standard).
- $(\mathcal{P}(\lambda \times t))_{t \geq 0}$ le semi-groupe du processus de Poisson d'intensité $\lambda \geq 0$.
- $(\text{Cauchy}(a \times t))_{t \geq 0}$ le semi-groupe de Cauchy de paramètre $a \geq 0$.
- $(\Gamma(\lambda, a \times t))_{t \geq 0}$ le semi-groupe des lois Gamma de paramètres $\lambda \geq 0$ et $a \times t \geq 0$. Rappelons qu'une loi Gamma de paramètres $\lambda > 0$ et $k > 0$ est la loi absolument continue sur \mathbb{R} admettant pour densité

$$p(x) = \mathbb{1}_{[0, +\infty[}(x) \frac{\lambda^k}{(k-1)!} x^{k-1} e^{-\lambda x}, \quad x \in \mathbb{R}.$$

Si X est une variable aléatoire de loi $\Gamma(\lambda, k)$, alors

$$\mathbb{E}[X] = \frac{k}{\lambda}, \quad \text{Var}(X) = \frac{k}{\lambda^2}, \quad \varphi_X(\theta) = \left(\frac{\lambda}{\lambda - i\theta}\right)^k.$$

L'extension du calcul de la fonction caractéristique pour k non entier est un peu délicat. Néanmoins on constate avec cette expression l'aspect multiplicatif en k de ces fonctions caractéristiques et ainsi la nature de semi-groupe de convolution.

• etc.

⊗ Identifions assez formellement la dérive :

$$\frac{d}{d\theta} \exp(tg(\theta)) \Big|_{\theta=0} = tg'(0) \exp(tg(0)) = tg'(0) = iat + t \int_{|x|<\infty} (ix e^{i0x} - ix) \nu(dx) = iat.$$

Elle est donc bien égale à a .

2. Simulation de processus à accroissements indépendants homogènes

2.1. MOUVEMENT BROWNIEN

⊗ La vitesse d'animation dépend énormément de la version du logiciel ou de l'ordinateur. Sur notre station de travail, l'usage de `xpause()` n'est pas nécessaire, ce qui ne l'a pas été en revanche avec les machines des étudiants. L'animation ne présente pas de fluctuations trop brutales et ne remet donc pas en cause la continuité connue des trajectoires browniennes.

⊗ Sans commentaire.

// question 2

N = 100; XT = [];

// on ne va pas créer un tableau $n \times N$ qui pourrait être trop volumineux

for i = 1:N;

 dX = grand(n, 1, "nor", 0, sqrt(T/n));

 XT = [XT; sum([0; dX])]; // valeur terminales d'une trajectoire issue de 0

end

function k = sturges(n); k = ceil(1+3.22*log10(n)); endfunction

scf(1); clf(); histplot(sturges(n), XT);

⊗

// question 3

getf("/home/phan/Scilab-Maple/kolmogorov.sci");

getf("/home/phan/Scilab-Maple/goodness-of-fit.sci");

deff("p = F(x)", "p = cdfnor('PQ', x, 0, sqrt(T))");

scf(2); cgofstest(XT, F);

Évidemment l'adéquation est bonne. Les seuls défauts peuvent venir du générateur puisque théoriquement la valeur terminale obtenue est de loi $\mathcal{N}(0, T)$.

⊗ Pour un paramètre de dérive égal à a et un coefficient de diffusion égal à σ , les accroissements sur un intervalle de longueur dt ont pour loi $\mathcal{N}(a dt, \sigma^2 dt)$. Soit

// trois trajectoires issue de 0

```

a = [0; 1; 2]; sigma = [1; 2; 3]; X = zeros(n+1, 3);
for i = 1:3;
    dX = grand(n, 1, "nor", a(i)*T/n, sigma(i)*sqrt(T/n));
    X(:,i) = cumsum([0; dX]);
end
scf(3); clf(); plot2d(time, X);
legend(["trajectoire 1", "trajectoire 2", "trajectoire 3"], 2);
halt("taper [entree] pour continuer\n");

```

2.2. PROCESSUS DE POISSON

Tout d'abord, rappelons que si μ_λ et $\mu_{\lambda'}$ sont deux lois de Poisson de paramètres respectifs λ et λ' , alors $\mu_\lambda * \mu_{\lambda'} = \mu_{\lambda+\lambda'}$ (la somme de deux variables aléatoires indépendantes de lois de Poisson de paramètres respectifs λ et λ' a pour loi la loi de Poisson de paramètre $\lambda + \lambda'$), ce qui se voit facilement sur les fonctions caractéristiques : pour tout $\theta \in \mathbb{R}$,

$$\varphi_\lambda(\theta) \times \varphi_{\lambda'}(\theta) = \exp(\lambda \times (e^{i\theta} - 1)) \times \exp(\lambda' \times (e^{i\theta} - 1)) = \exp((\lambda + \lambda') \times (e^{i\theta} - 1)) = \varphi_{\lambda+\lambda'}(\theta).$$

Le processus de Poisson d'intensité λ est donc associé au semi-groupe $(\mathcal{P}(\lambda t))_{t \geq 0}$ de convolution et les accroissements sur un intervalle de longueur dt ont pour loi $\mathcal{P}(\lambda \times dt)$.

⊗ Il faut prendre λ tel que $T \times \lambda$ soit assez grand pour voir qu'il se passe quelque chose.

```
// Processus de Poisson
```

```

lambda = 10;
dX = grand(n, 1, "poi", T*lambda/n);
X = cumsum([0; dX]); // trajectoire issue de 0
scf(0); clf(); plot2d(time, X);
XT = [];
// on ne va pas créer un tableau n\times N qui pourrait être trop volumineux
for i = 1:N;
    dX = grand(n, 1, "poi", T*lambda/n);
    XT = [XT; sum([0; dX])]; // valeur terminales d'une trajectoire issue de 0
end

```

⊗ Les trajectoires sont croissantes et visiblement en escaliers.

⊗ Il y a effectivement une dérive puisque les trajectoires sont croissantes. La pente globale sur $[0, T]$ semble de l'ordre de λ . Plus précisément, la fonction caractéristique d'une loi de Poisson de paramètre λt est $\varphi(\theta) = \exp(\lambda t(e^{i\theta} - 1))$ et donc la fonction g de la formule de Lévy–Khinchine est dans ce cas

$$g(\theta) = \lambda(e^{i\theta} - 1) = \lambda \sum_{n \geq 1} \frac{(i\theta)^n}{n!}.$$

La dérive se lit sur le terme d'ordre 1, c'est $a = \lambda$ comme attendu.

Évidemment $(X_t - at)_{t \geq 0}$ est une martingale, car X est un processus à accroissements indépendants et $at = \mathbb{E}[X_t]$.

⊗ Le plus dur est d'arrêter une liste de valeurs possibles pour faire le test du χ^2 .

```

ptheo = exp(-lambda*T); // premi\`ere probabilit\`e
M = 0; // valeur maximale;

// les premi\`eres probabilit\`es sont peut-\`etre trop faibles
// alors on cumule...

while N*ptheo(M+1) < 5;
    M = M+1; xtheo = M; ptheo = ptheo+ptheo*lambda*T/M;
end

xtheo = M; // premiere valeur raisonnable

while N*ptheo(M+1) >= 5;
    M = M+1; xtheo = [xtheo; M]; ptheo = [ptheo; ptheo(M)*lambda*T/M];
end

ptheo(M+1) = 1-sum(ptheo(1:M)); // reste de la loi de Poisson
XT = min(max(XT, xtheo(1)), M); // troncations inf\`erieure et sup\`erieure
scf(1); dgoftest(XT, xtheo, ptheo); // graphiques et test du chi^2
halt("taper [entree] pour continuer\n");

```

2.3. PROCESSUS DE CAUCHY

Tout d'abord, rappelons que si μ_a et $\mu_{a'}$ sont deux lois de Cauchy de paramètres respectifs $a \geq 0$ et $a' \geq 0$, alors $\mu_a * \mu_{a'} = \mu_{a+a'}$ (la somme de deux variables aléatoires indépendantes de lois de Cauchy de paramètres respectifs a et a' a pour loi la loi de Cauchy de paramètre $a + a'$), ce qui se voit facilement sur les fonctions caractéristiques :

$$\varphi_a(\theta) \times \varphi_{a'}(\theta) = e^{a|\theta|} \times e^{a'|\theta|} = e^{(a+a')|\theta|} = \varphi_{a+a'}(\theta), \quad \theta \in \mathbb{R}.$$

Le processus de Cauchy d'intensité a est donc associé au semi-groupe $(\mathcal{C}(at))_{t \geq 0}$ de convolution et les accroissements sur un intervalle de longueur dt ont pour loi $\mathcal{C}(a \times dt)$.

⊞ Soit $a > 0$. La fonction de répartition de la loi de Cauchy de paramètre a est

$$F_a(x) = \int_{-\infty}^x \frac{a/\pi}{a^2 + y^2} dy = \int_{-\infty}^x \frac{1/\pi}{1 + (y/a)^2} d(y/a) = \int_{-\infty}^{x/a} \frac{1/\pi}{1 + z^2} dz = F_1(x/a), \quad x \in \mathbb{R}.$$

et

$$\begin{aligned} F_1(x) &= \frac{1}{\pi} \int_{-\infty}^x \frac{1}{1 + y^2} dy = \frac{1}{\pi} [\arctan(y)]_{-\infty}^x = \frac{1}{\pi} (\arctan(x) - \arctan(-\infty)) \\ &= \frac{1}{\pi} (\arctan(x) - (-\pi/2)) = \frac{1}{\pi} \arctan(x) + \frac{1}{2}. \end{aligned}$$

D'où, $F_a(x) = \arctan(x/a)/\pi + 1/2$. Ces fonctions de répartition sont des homéomorphismes de \mathbb{R} dans $]0, 1[$, on peut donc simuler la loi de Cauchy de paramètre a par inversion de la fonction de répartition : si U est de loi uniforme sur $[0, 1]$, $X = F_a^{-1}(U)$ a pour loi la loi de Cauchy de paramètre a . Explicitement,

$$\begin{aligned} x = F_a^{-1}(u) &\iff F_a(x) = u \iff \arctan(x/a)/\pi + 1/2 = u \\ &\iff \arctan(x/a) = \pi(u - 1/2) \iff x = a \tan(\pi(u - 1/2)). \end{aligned}$$

Nous pouvons donc définir

```
mprintf("Processus de Cauchy\n");
```



```
function x = cauchydeviate(n, m, a);
    x = a*tan(%pi*(grand(n, m, "def")-0.5));
endfunction
```

en notant que `tan` est « vectorialisée ».

☞ Nous prenons $a = 1$.

```
a = 1;
dX = cauchydeviate(n, 1, a*T/n);
X = cumsum([0; dX]); // trajectoire issue de 0
scf(0); clf(); plot2d(time, X);
```

Les trajectoires présentent de brusques variations. Elles comportent certainement des sauts, mais ne sont pas en escaliers, elles ont donc une partie continue non nulle.

☞ Pour l'adéquation, nous procédons au test de Kolmogorov–Smirnov.

```
XT = [];
// on ne va pas cr\eer un tableau n\times N qui pourrait \^etre trop volumineux
for i = 1:N;
    dX = cauchydeviate(n, 1, a*T/n);
    XT = [XT; sum([0; dX])]; // valeur terminales d'une trajectoire issue de 0
end
deff("p = F(x)", "p = atan(x/(a*T))/%pi+0.5");
scf(2); cgoftest(XT, F);
halt("taper [entree] pour continuer\n");
```

2.4. PROCESSUS GAMMA

Tout d'abord, rappelons que si $\mu_{a,\lambda}$ et $\mu_{a',\lambda}$ sont deux lois Gamma de paramètres respectifs (a, λ) et (a', λ) , alors $\mu_a * \mu_{a'} = \mu_{a+a'}$ (la somme de deux variables aléatoires indépendantes de lois Gamma de paramètres respectifs (a, λ) et (a', λ) a pour loi la loi Gamma de paramètre $(a + a', \lambda)$), ce qui se voit facilement sur les fonctions caractéristiques : pour tout $\theta \in \mathbb{R}$,

$$\varphi_{a,\lambda}(\theta) \times \varphi_{a',\lambda}(\theta) = \left(\frac{\lambda}{\lambda - i\theta}\right)^a \times \left(\frac{\lambda}{\lambda - i\theta}\right)^{a'} = \left(\frac{\lambda}{\lambda - i\theta}\right)^{a+a'} = \varphi_{a+a',\lambda}(\theta).$$

Le processus Gamma de paramètres (a, λ) est donc associé au semi-groupe de convolution $(\Gamma(at, \lambda))_{t \geq 0}$ et les accroissements sur un intervalle de longueur dt ont pour loi $\Gamma(a \times dt, \lambda)$.

☞ Sans commentaire ou presque.

```
mprintf("Processus Gamma\n");
a = 8; lambda = 10;
dX = grand(n, 1, "gam", a*T/n, lambda);
X = cumsum([0; dX]); // trajectoire issue de 0
scf(0); clf(); plot2d(time, X);
XT = [];
// on ne va pas cr\eer un tableau n\times N qui pourrait \^etre trop volumineux
```

```

for i = 1:N;
    dX = grand(n, 1, "gam", a*T/n, lambda);
    XT = [XT; sum([0; dX])]; // valeur terminales d'une trajectoire issue de 0
end

deff("p = F(x)", "p = cdfgam('PQ', x, a*T, lambda)");
scf(2); cgoftest(XT, F);
halt("taper [entree] pour continuer\n");

```

Les trajectoires sont croissantes, comportent une partie à sauts et une partie continue.

Annexe : calculs de fonctions caractéristiques

2.1. LOIS NORMALES

Soient $m \in \mathbb{R}$ et $\sigma > 0$. La fonction caractéristique de la loi normale de moyenne m et d'écart-type σ est

$$\theta \in \mathbb{R} \longmapsto \exp(im\theta - \sigma^2\theta^2/2).$$

Démonstration. — Soit X une variable aléatoire réelle de loi $\mathcal{N}(m, \sigma^2)$. Nous savons alors que $Z = (X - m)/\sigma$ est une variable aléatoire réelle de loi $\mathcal{N}(0, 1)$. On a, pour tout $\theta \in \mathbb{R}$,

$$\begin{aligned} \varphi_X(\theta) &= \mathbb{E}[e^{i\theta X}] = \mathbb{E}[e^{i\theta(X-m)} \times e^{i\theta m}] = \mathbb{E}[e^{i\sigma\theta(X-m)/\sigma} \times e^{i\theta m}] \\ &= \mathbb{E}[e^{i\sigma\theta(X-m)/\sigma}] \times e^{i\theta m} = \mathbb{E}[e^{i\sigma\theta Z}] \times e^{i\theta m} = \varphi_Z(\sigma\theta) \times e^{i\theta m}. \end{aligned}$$

Il nous suffit donc de calculer la fonction caractéristique φ_Z de la loi normale $\mathcal{N}(0, 1)$. Nous avons

$$\varphi_Z(\theta) = \int_{-\infty}^{+\infty} e^{i\theta z} e^{-z^2/2} \frac{dz}{\sqrt{2\pi}}.$$

Démontrer que φ_Z est de classe C^1 et que sa fonction dérivée s'obtient par dérivation sous le signe somme est classique, nous nous en dispensons. Nous avons donc

$$\begin{aligned} \varphi'_Z(\theta) &= \frac{d}{d\theta} \left(\int_{-\infty}^{+\infty} e^{i\theta z} e^{-z^2/2} \frac{dz}{\sqrt{2\pi}} \right) = \int_{-\infty}^{+\infty} \frac{\partial}{\partial \theta} (e^{i\theta z} e^{-z^2/2}) \frac{dz}{\sqrt{2\pi}} \\ &= \int_{-\infty}^{+\infty} \frac{\partial}{\partial \theta} (e^{i\theta z}) e^{-z^2/2} \frac{dz}{\sqrt{2\pi}} = \int_{-\infty}^{+\infty} iz e^{i\theta z} e^{-z^2/2} \frac{dz}{\sqrt{2\pi}}. \end{aligned}$$

Puis nous intégrons par parties en la variable z en notant que $(e^{-z^2/2})' = -z e^{-z^2/2}$

$$\begin{aligned} \varphi'_Z(\theta) &= \int_{-\infty}^{+\infty} -i e^{i\theta z} \frac{d}{dz} (e^{-z^2/2}) \frac{dz}{\sqrt{2\pi}} = \left[-i e^{i\theta z} e^{-z^2/2} \frac{1}{\sqrt{2\pi}} \right]_{-\infty}^{+\infty} \\ &\quad + \int_{-\infty}^{+\infty} i^2 \theta e^{i\theta z} e^{-z^2/2} \frac{dz}{\sqrt{2\pi}} = 0 - \theta \int_{-\infty}^{+\infty} e^{i\theta z} e^{-z^2/2} \frac{dz}{\sqrt{2\pi}} = -\theta \times \varphi_Z(\theta). \end{aligned}$$

Ainsi la fonction φ_Z vérifie l'équation différentielle linéaire homogène à coefficients lipschitziens $\varphi' = -\theta \times \varphi$ sur \mathbb{R} et la condition initiale $\varphi_Z(0) = 1$. Comme la fonction $\theta \mapsto e^{-\theta^2/2}$ vérifie aussi ces conditions, par unicité, on a $\varphi_Z(\theta) = e^{-\theta^2/2}$ pour tout $\theta \in \mathbb{R}$. La formule annoncée dans le lemme s'en déduit immédiatement : pour $\theta \in \mathbb{R}$,

$$\begin{aligned} \varphi_X(\theta) &= \mathbb{E}[e^{i\theta X}] = \mathbb{E}[e^{im\theta} \times e^{i\theta(X-m)}] = e^{im\theta} \times \mathbb{E}[e^{i\theta(X-m)}] \\ &= e^{im\theta} \times \mathbb{E}[e^{i(\sigma\theta)(X-m)/\sigma}] = e^{im\theta} \times \mathbb{E}[e^{i(\sigma\theta)Z}] = e^{im\theta} \times \varphi_Z(\sigma\theta) = \exp(im\theta - \sigma^2\theta^2/2), \end{aligned}$$

ce qui était annoncé. □

2.2. LOIS DE POISSON

Soit $\lambda \geq 0$. La fonction caractéristique de la loi de Poisson de paramètre λ est

$$\theta \in \mathbb{R} \longmapsto \exp(\lambda \times (e^{i\theta} - 1)).$$

Démonstration. — C'est du calcul élémentaire. Si X est une variable aléatoire de loi la loi de Poisson de paramètre λ , on a

$$\varphi_X(\theta) = \mathbb{E}[e^{i\theta X}] = \sum_{n \geq 0} e^{i\theta n} e^{-\lambda} \frac{\lambda^n}{n!} = e^{-\lambda} \sum_{n \geq 0} \frac{(\lambda e^{i\theta})^n}{n!} = e^{-\lambda} \exp(\lambda e^{i\theta}) = \exp(\lambda \times (e^{i\theta} - 1)),$$

comme annoncé. \square

2.3. LOIS DE CAUCHY

Soit $a \geq 0$. La fonction caractéristique de la loi de Cauchy de paramètre a est

$$\theta \in \mathbb{R} \longmapsto \exp(a|\theta|).$$

Démonstration. — Prenons $a = 1$ et soit $\theta \in \mathbb{R}$. Nous allons calculer

$$\varphi(\theta) = \int_{\mathbb{R}} \frac{e^{i\theta x}}{1+x^2} \frac{dx}{\pi} = \lim_{r \rightarrow \infty} \int_r^{-r} \frac{e^{i\theta x}}{1+x^2} \frac{dx}{\pi} = \lim_{r \rightarrow \infty} I(r)$$


à l'aide d'outils d'analyse complexe, en particulier le théorème des résidus. Nous avons

$$f_\theta(z) = \frac{1}{\pi} \frac{e^{i\theta z}}{1+z^2} = \frac{e^{i\theta z}}{\pi} \left(\frac{i/2}{z+i} - \frac{i/2}{z-i} \right) = \frac{i e^{i\theta z}}{2\pi} \left(\frac{1}{z+i} - \frac{1}{z-i} \right)$$

qui est holomorphe sur $\mathbb{C} \setminus \{-i, i\}$, de pôles $\{-i, i\}$ et de résidus

$$\text{Res}(f_\theta, -i) = \frac{i e^{i\theta \times (-i)}}{2\pi} = \frac{i e^\theta}{2\pi}, \quad \text{Res}(f_\theta, i) = -\frac{i e^{i\theta \times i}}{2\pi} = -\frac{i e^{-\theta}}{2\pi}.$$

Pour $r > 1$, considérons



$$\begin{cases} I_+(r) = \int_{C_+} f_\theta(z) dz = 2i\pi \text{Res}(f_\theta, i) = e^{-\theta}, \\ I_-(r) = \int_{C_-} f_\theta(z) dz = 2i\pi \text{Res}(f_\theta, -i) = -e^\theta. \end{cases}$$

Par passage en coordonnées polaires, on a

$$e^{-\theta} = I_+(r) = I(r) + \int_0^\pi f_\theta(r e^{i\alpha}) \times r d\alpha.$$

Ainsi,

$$\begin{aligned} |e^{-\theta} - I(r)| &\leq \int_0^\pi |f_\theta(r e^{i\alpha})| \times r d\alpha = \int_0^\pi \frac{\exp(-\theta r \sin \alpha)}{|1 + r^2 \exp(2i\alpha)|} \times r d\alpha \\ &= \int_0^\pi \frac{\exp(-\theta r \sin \alpha)}{|1/r + r \exp(2i\alpha)|} d\alpha \leq \int_0^\pi \frac{\exp(-\theta r \sin \alpha)}{r - 1/r} d\alpha. \end{aligned}$$

Comme pour $\alpha \in [0, \pi]$, $\sin \alpha \geq 0$, et ainsi, si $\theta \geq 0$, $\exp(-\theta r \sin \alpha) \leq 1$. Pour $\theta \geq 0$, on a alors

$$|e^{-\theta} - I(r)| \leq \frac{\pi}{r - 1/r} \longrightarrow 0 \quad \text{quand } r \rightarrow \infty,$$

et ainsi $\varphi(\theta) = e^{-\theta}$ pour $\theta \geq 0$. Lorsque $\theta \leq 0$, on peut considérer $|e^\theta - I(r)| = |I(r) + I_-(r)|$ et montrer comme précédemment que ça tend vers 0 quand r tend vers l'infini et qu'ainsi

$\varphi(\theta) = e^\theta$ pour $\theta \leq 0$, ou simplement remarquer que puisque la densité de la loi de Cauchy est paire, alors la fonction caractéristique est elle aussi paire, et donc, de $\varphi(\theta) = e^{-\theta}$ pour $\theta \geq 0$, en déduire que $\varphi(\theta) = e^{-|\theta|}$ pour tout $\theta \in \mathbb{R}$.

Pour une loi de Cauchy de paramètre $a > 0$, nous savons que si X est de loi $\mathcal{C}(a)$, alors X/a est de loi $\mathcal{C}(1)$. Ainsi $\varphi_a(\theta) = \mathbb{E}[e^{i\theta X}] = \mathbb{E}[e^{ia\theta(X/a)}] = \varphi(a\theta) = e^{-a|\theta|}$. \square

2.4. LOIS GAMMA

Soient a et λ deux réels strictement positifs. La fonction caractéristique de la loi Gamma de paramètres (a, λ) est

$$\theta \in \mathbb{R} \longmapsto \exp\left(\frac{\lambda}{\lambda - i\theta}\right)^a.$$

Démonstration. — Soient $a > 0$ et $\lambda > 0$. Posons

$$\varphi(\theta) = \int_0^\infty e^{i\theta x} \frac{\lambda^a}{\Gamma(a)} x^{a-1} e^{-\lambda x} dx = \int_0^\infty C x^{a-1} e^{(i\theta-\lambda)x} dx.$$

Cette fonction est de classe C^1 et on peut dériver sous le signe somme

$$\varphi'(\theta) = \frac{d}{d\theta} \int_0^\infty C x^{a-1} e^{(i\theta-\lambda)x} dx = \int_0^\infty C \frac{\partial}{\partial \theta} (x^{a-1} e^{(i\theta-\lambda)x}) dx = \int_0^\infty iC x^a e^{(i\theta-\lambda)x} dx$$

Puis, en intégrant par parties,

$$\varphi'(\theta) = \frac{i}{i\theta - \lambda} [C x^a e^{(i\theta-\lambda)x}]_0^\infty - \frac{ia}{i\theta - \lambda} \int_0^\infty iC x^{a-1} e^{(i\theta-\lambda)x} dx = 0 + \frac{ia}{\lambda - i\theta} \varphi(\theta).$$

Non rigoureusement,

$$\frac{\varphi'(\theta)}{\varphi(\theta)} = -a \frac{-i}{\lambda - i\theta} \implies \ln(\varphi(\theta)) = -a \ln(\lambda - i\theta) + \text{Cte} \implies \varphi(\theta) = \text{Cte} \times (\lambda - i\theta)^{-a}$$

et puisque $\varphi(0) = 1 = \text{Cte} \times \lambda^{-a}$, $\text{Cte} = \lambda^a$, et on a

$$\varphi(\theta) = \left(\frac{\lambda}{\lambda - i\theta}\right)^a.$$

Plus rigoureusement, φ vérifie une équation différentielle du premier ordre à coefficients lipschitziens puisque le facteur $ia/(\lambda - i\theta)$ est borné dans \mathbb{C} pour $\lambda > 0$ fixé et $\theta \in \mathbb{R}$. Cette équation différentielle admet donc une unique solution qui est de plus définie sur \mathbb{R} . Notons que pour $\theta \in \mathbb{R}$, $\lambda/(\lambda - i\theta) \in \{z \in \mathbb{C} : \text{Re } z > 0\}$. Considérons la détermination principale du logarithme $\text{Log} : \mathbb{C} \setminus \mathbb{R}_- \rightarrow \mathbb{C}$ et la fonction

$$\phi : \theta \in \mathbb{R} \longmapsto \left(\frac{\lambda}{\lambda - i\theta}\right)^a = \exp\left(a \text{Log}\left(\frac{\lambda}{\lambda - i\theta}\right)\right) \in \mathbb{C}$$

qui est bien définie, dérivable, vérifiant l'équation différentielle qui nous occupe et satisfaisant $\phi(0) = 1$. Par unicité, on a donc $\varphi = \phi$. \square

COMPLÉMENTS AUX TRAVAUX PRATIQUES N° 1

1. Retour sur le cas poissonnien

Certains auteurs présentent les lois de Poisson (Siméon Denis) comme apparaissant à travers le résultat asymptotique connu sous le nom de « loi des événements rares » ou « loi des petits nombres ». Celui-ci affirme que si $(p_n)_{n \geq 0}$ est une suite de réels positifs (inférieurs ou égaux à 1) tels que $n \times p_n \rightarrow \lambda \geq 0$ lorsque n tend vers l'infini, alors la suite des lois binomiales $\mathcal{B}(n, p_n)$ converge vers la loi de Poisson $\mathcal{P}(\lambda)$ de paramètre λ . Ce résultat permet, pour $\lambda \geq 0$ fixé, de simuler de manière approchée le tirage selon la loi de Poisson de paramètre λ à l'aide d'un schéma de Bernoulli de paramètres n et λ/n .

⊞ Considérer chaque pas de temps comme une épreuve du schéma de Bernoulli destiné à approcher la loi de Poisson de paramètre $\lambda = 1$ au temps $T = 1$. Représenter la chronologie du nombre de succès et donc une trajectoire approchée d'un tel processus de Poisson.

2. Retour sur le cas brownien

Si la loi des événements rares permet d'approcher un processus de Poisson via un schéma de Bernoulli, une version étendue du théorème central limite permet une démarche semblable pour le mouvement brownien.

Le « théorème d'invariance de Donsker » affirme qu'on peut approcher le mouvement brownien (unidimensionnel) par toute marche aléatoire raisonnable convenablement normalisée. Sans rentrer dans les détails : Soit $(\xi_n)_{n \geq 0}$ une suite de variables aléatoires réelles indépendantes, identiquement distribuées de moyenne nulle et de variance 1. En posant, pour $k \geq 0$, $X_{k/n}^{(n)} = \sum_{\ell=1}^k \xi_\ell / \sqrt{n}$ et en prolongeant cette suite de manière continue et affine par morceaux de la manière la plus simple qu'il soit, on obtient un processus $X^{(n)} = (X_t^{(n)})_{t \geq 0}$ approchant le mouvement brownien standard.

⊞ Mettre en œuvre ce schéma d'approximation en partant de la marche aléatoire simple symétrique sur \mathbb{Z} pour représenter une ou des trajectoires de $X^{(n)} = (X_t^{(n)})_{0 \leq t \leq T}$ (on pourra prendre $T = 1$ et n raisonnablement grand).

⊞ Comparer les lois terminales : par simulation d'un nombre N de valeurs observées de $X_T^{(n)}$ avec la loi normale $\mathcal{N}(0, T)$; et théoriquement en explicitant la loi de $X_T^{(n)}$.

Remarque. — L'utilisation d'un schéma de simulation apparemment grossier (précédemment on aura effectué avec le tirage d'accroissements gaussiens des simulations plus fidèles) comme celui-ci est très souvent pertinent pour les simulations de solutions d'équations différentielles stochastiques ou pour simuler le mouvement brownien sur un domaine borné (voir théorème de Kakutani) ou sur une surface ou variété riemannienne.

On peut aussi combiner les méthodes (loi des événements rares, théorème d'invariance de Donsker) pour simuler des processus à sauts plus généraux (processus de Lévy, de Markov, solutions d'équations différentielles stochastiques, ...).

COMPLÉMENTS AUX TRAVAUX PRATIQUES N° 1. — ÉLÉMENTS D'EXPLICATIONS

1. Retour sur le cas poissonnien

☞ Pour la simulation d'une trajectoire d'un processus de Poisson par la loi des événements rares, on a

```
// Retour sur le cas poissonnien  
  
T = 1; n = 1000; lambda = 1;  
time = (T/n)*[0:n]';  
Trials = grand(n, 1, "bin", 1, lambda*T/n);  
X = cumsum([0; Trials], "r");  
clf(); plot2d(time, X);  
xtitle("Processus de Poisson et loi des evenements rares");  
halt("*** taper [entree] pour continuer ***\n");
```

Allons un peu plus loin en simulant N trajectoires pour ne conserver que leurs valeurs terminales (il y a du gaspillage de mémoire vive).

```
N = 1000;  
Trials = grand(n, N, "bin", 1, lambda*T/n);  
// X = cumsum([zeros(1, N); Trials], "r"); // inutile d'avoir les N trajectoires  
// XT = X(n+1, :)'; // puisque seules les N valeurs terminales servent  
XT = sum(Trials, "r")';
```

Nous avons à comparer la distribution observée de la variable terminale avec la loi de Poisson $\mathcal{P}(\lambda T)$ avec le test du χ^2 d'adéquation. Nous limiterons de manière artificielle et maladroite aux valeurs entières comprises entre 0 et la valeur maximale observée x_{\max} . La distribution observée sera alors comparée avec la distribution de Poisson tronquée en x_{\max} :

$$\pi\{k\} = e^{-\lambda T} \frac{(\lambda T)^k}{k!} \quad \text{pour } 0 \leq k < x_{\max}, \quad \text{et} \quad \pi\{x_{\max}\} = 1 - (\pi\{0\} + \cdots + \pi\{x_{\max} - 1\}).$$

Soit

```
xmax = max(XT);  
xtheo = [0:xmax]'; // une colonne  
ptheo = zeros(xmax+1, 1); // initialisation  
ptheo(1) = exp(-lambda*T); // premier terme  
s = 0;  
for k = 1:xmax;  
    ptheo(k+1) = ptheo(k)*lambda*T/k; // attention aux indices  
    s = s+ptheo(k);  
end  
ptheo(xmax+1) = 1-s; // reste de la loi de Poisson  
// on \`ecrase au passage le ptheo(xmax+1) d'\`efini au dernier tour de boucle
```

```
dgofstest(XT, xtheo, ptheo); // graphiques et test du chi^2
halt("*** taper [entree] pour continuer ***\n");
```

Le calcul des probabilités successives aura été fait en remarquant que

$$\pi\{0\} = e^{-\lambda T} \quad \text{et} \quad \pi\{k\} = \pi\{k-1\} \times \frac{(\lambda T)^k}{k} \quad \text{pour } k \geq 1,$$

et en prenant garde à l'indexation de SCILAB : $\text{ptheo}(\langle k \rangle + 1) = \pi\{k\}$.

☞ Nous rappelons que si $U : (\Omega, \mathcal{A}, \mathbb{P}) \rightarrow [0, 1]$ est une variable aléatoire de loi uniforme sur $[0, 1]$ (on supposera la valeur 1 strictement impossible), alors pour $\lambda > 0$, $T = -\ln(1 - U)/\lambda$ est une variable aléatoire réelle positive de loi exponentielle de paramètre λ .

☞ Pour simuler une variable de loi de Poisson de paramètre λ , on peut simuler une file d'attente à temps de record de paramètre λ et compter le nombre de passages au temps $T = 1$ ou encore une file d'attente de temps de record de paramètre 1 observée au temps λT .

```
function n = Poissonrand(lambda)
    n = -1;
    while lambda >= 0;
        n = n+1;
        lambda = lambda+log(1-grand(1, 1, "def"));
    end
endfunction
```

Si on veut conserver la trace des temps de record ou des dates de saut dans $[0, 1]$, on pourra taper

```
function [n, scores] = Poissonrand(lambda)
    s = 0; scores = [];
    while s <= 1;
        s = s-log(1-grand(1, 1, "def"))/lambda;
        scores = [scores; s];
    end
    n = size(scores, 1)-1;
    scores(n+1) = 1; // ignorer ce qu'il se passe apr\`es $t = 1$
endfunction
```

☞ Nous ne tracerons pas les trajectoires implicitement ou non définies par la chaque appel de la fonction `Poissonrand`. Regardons simplement ce qu'il se passe au niveau de la variable terminale.

```
clear n lambda T X XT ptheo xtheo;

N = 1000; lambda = 1; T = 1;
for k = 1:N; X(k) = Poissonrand(lambda*T); end // X est un vecteur colonne
xmax = max(X); xtheo = [0:xmax]';
ptheo = zeros(xmax+1, 1); ptheo(1) = exp(-lambda*T); s = 0;
for k = 1:xmax;
    ptheo(k+1) = ptheo(k)*lambda*T/k;
    s = s+ptheo(k);
end
ptheo(xmax+1) = 1-s; // reste de la loi de Poisson

dgofstest(X, xtheo, ptheo); // graphiques et test du chi^2
halt("*** taper [entree] pour continuer ***\n");
```

Remarque. — Le choix de x_{\max} aurait dû être fait plus judicieusement dans tout ce qui précède. Nous aurions pu le choisir de sorte qu'il soit le plus grand entier k tel que $N \times \pi\{k\} \geq 5$ (la valeur 5 provenant des condition d'applicabilité du test [asymptotique] du χ^2). Par exemple le code précédent aurait été écrit

```
N = 1000; lambda = 1; T = 1;
clear ptheo; ptheo(1) = exp(-lambda*T); s = 0; xmax = 0;
while N*ptheo(xmax+1) >= 5;
    xmax = xmax+1;
    ptheo(xmax+1) = ptheo(xmax)*lambda*T/xmax;
    s = s+ptheo(xmax);
end
ptheo(xmax+1) = 1-s; xtheo = [0:xmax]
for k = 1:N; X(k) = min(Poissonrand(lambda*T), xmax); end
dgoftest(X, xtheo, ptheo);
halt("*** taper [entree] pour continuer ***\n");
```

2. Retour sur le cas brownien

⊞ Considérons la marche aléatoire simple sur \mathbb{Z} . Ses accroissements sont indépendants de loi $(\delta_{\{-1\}} + \delta_{\{1\}})/2$ de moyenne 0 et de variance 1.

```
// Retour sur le cas brownien
clear();
getf("/home/phan/Scilab-Maple/kolmogorov.sci");
getf("/home/phan/Scilab-Maple/goodness-of-fit.sci");
T = 1; n = 1000; time = (T/n)*[0:n]';
dX = sqrt(T/n)*(2*grand(n, 1, "bin", 1, 0.5));
X = cumsum([0; dX], "r");
clf(); plot2d(time, X);
halt("*** taper [entree] pour continuer ***\n");
```

⊞ Les accroissements générés par la machine sont supposés avoir une propriété statistique similaire à l'indépendance. Notamment les théorèmes asymptotiques classiques doivent pouvoir être approximativement vérifiés (cela fait partie des tests pratiqués par les spécialistes des générateurs de nombres aléatoires). Nous devons donc avoir un théorème central limite. Vérifions-le avec le test de Kolmogorov–Smirnov sur $N = 100$ valeurs terminales (procédure `cgoftest` définie dans `goodness-of-fit.sci`).

```
N = 100; // 1000 serait peut-être trop pour kolmogorovcdf(.)
dX = sqrt(T/n)*(2*grand(n, N, "bin", 1, 0.5));
// X = cumsum([zeros(1, N); dX], "r"); // inutile d'avoir les N trajectoires
// XT = X(n+1,:)' ; // puisque seules les N valeurs terminales servent
XT = sum(dX, "r")';
deff("y = F(x)", "y = cdfnor(\"PQ\", x, 0, sqrt(T))"); cgoftest(XT, F);
halt("*** taper [entree] pour continuer ***\n");
```


NON PROPOSÉ


1. L'escalier de Cantor

L'escalier de Cantor est généralement présentée en cours d'intégration de L3 conjointement à l'ensemble triadique C de Cantor. Ce dernier est obtenu en partant de l'intervalle $C_0 = [0, 1]$ à qui on enlève dans un premier temps l'intervalle ouvert $]1/3, 2/3[$ pour obtenir C_1 ; puis des intervalles restants, on enlève les tiers ouverts centraux pour obtenir C_2 , etc. On obtient ainsi une suite décroissante de compacts $(C_n)_{n \geq 0}$ de longueur (de mesure de Lebesgue) respective $(2/3)^n$. L'ensemble $C = \bigcap_{n \geq 0} C_n$ est alors un compact non vide de mesure nulle. Il est sans point isolé et de cardinal continu.

% L'escalier de Cantor


```
def cantorset(expr u,v,depth) =
  gdraw (u,depth)..(v,depth);
  if depth > 0:
    cantorset(u,1/3[u,v],depth-1);
    cantorset(2/3[u,v],v,depth-1);
  fi
enddef;
```

```
beginfig(thisfig);
  options noframe;
  setrange(0,1,12cm,0,10,1cm);
  pickup rule.nib;
  cantorset(0,1,vmax);
  xticks.top(vmax) "0", "1/9", "2/9", "1/3", "4/9", "5/9", "2/3", "7/9", "8/9", "1";
endfig;
```



Dans le même esprit, on construit une suite de fonctions continues, croissantes affines par morceaux valant 0 en 0 et 1 en 1, et convergeant uniformément sur \mathbb{R} vers une fonction continue F , croissante valant 0 en 0 et 1 en 1 et constante sur les composantes connexes de

C^c . Les graphes suivants précisent la construction de F .



Quant au code MetaPost suivant, il met l'accent la construction récursive de ces figures.

```
def cantorgraph(expr u, v, depth) =
  if depth > 0:
    cantorgraph(u, (xpart(1/3[u,v]), ypart(1/2[u,v])), depth-1)
    --cantorgraph((xpart(2/3[u,v]), ypart(1/2[u,v])), v, depth-1)
  else: u--v fi
enddef;

n := 5;
for i = 0 upto n:
  beginfig(thisfig);
  setrange(-1/6, 7/6, 8/3cm, 0, 1, 2cm);
  pickup rule.nib;
  gdraw (hmin, 0) -- cantorgraph((0, 0), (1, 1), i) -- (hmax, 1);
  yticks.lft(hmin) "0", "1/4", "1/2", "3/4", "1";
  xticks.bot(vmin) "0", "1/3", "2/3", "1";
  endfig;
endfor
```

1.1. LOI DE VARIABLE ALÉATOIRE

La fonction F est une fonction de répartition telle que $F(0) = 0$ et $F(1) = 1$. Il existe donc une unique mesure de probabilité μ sur $[0, 1]$ dont F est la fonction de répartition. Soit $X : (\Omega, \mathcal{A}, \mathbb{P}) \rightarrow [0, 1]$ une variable aléatoire de loi μ .

- ☞ Déterminer l'espérance de X par des arguments élémentaires mais rigoureux.
- ☞ Mettre en place en SCILAB une fonction permettant de calculer de manière approchée $\mathbb{E}[f(X)]$ pour toute fonction $f : [0, 1] \rightarrow \mathbb{R}$ continue. On pourra pour cela approcher X par des variables aléatoires de loi uniforme sur des ensembles finis à préciser.
- ☞ Approcher $\mathbb{E}[X^2]$ ainsi que la variance de X . Leur détermination théorique vous semble-t-elle accessible? Si oui, comment?

1.2. PROCESSUS À VARIATION FINIE

Considérons maintenant la fonction F comme un processus croissant (déterministe). C'est une fonction continue dont la dérivée est définie et nulle presque partout, pourtant en général,

$$\mu(\mathbb{1}_{]a,t]}g) =: \int_a^t g(s) dF(s) \neq \int_a^t g(s)F'(s) ds = 0$$

où F' est une fonction nulle en tout point où F est dérivable. (On notera la convention concernant les bornes d'intégration. Elle est nécessaire pour avoir une relation Chasles lorsque la mesure μ comporte une partie discrète, notamment ici si $\mu\{a\} > 0$, ce qui n'est cependant pas le cas ici.)

⊛ Définir avec SCILAB la fonction $t \in \mathbb{R} \mapsto F(t)$. L'utiliser cette définition dans l'implémentation d'un calcul approché par des sommes de Riemann

$$\int_0^t g(s) dF(s) \approx \sum_{0=t_0 < \dots < t_n=t} g(t_k)(F(t_{k+1}) - F(t_k))$$

prises sur des subdivisions régulières.

⊛ Écrire une fonction SCILAB permettant de calculer ces approximations d'intégrales de manière récursive.

⊛ Mettre en œuvre les deux méthodes pour $g(s) = s$ et $g(s) = s^2$ avec $t = 1/4, 1/2, 3/4, 1$ (les valeurs exactes ne sont pas demandées).

⊛ Quelle méthode semble la plus performante? Vous semble-t-il que l'on puisse aisément généraliser cette dernière méthode à toute fonction de répartition? voire plus généralement à toutes fonctions monotones ou différences de telles fonctions (fonctions à variation finie)?

Remarque. — L'évocation de la fonction de Cantor est une forme d'introduction au temps local brownien dont il sera question ultérieurement.

NON PROPOSÉ. — ÉLÉMENTS D'EXPLICATIONS

1. L'escalier de Cantor

1.1. LOI DE VARIABLE ALÉATOIRE

⊞ La loi de X est symétrique par rapport à $1/2$, comme X est bornée, elle est intégrable et alors $\mathbb{E}[X] = 1/2$.

⊞ Pour $n \geq 0$, posons $X_n^-(\omega) = k/3^n$ et $X_n^+(\omega) = (k+1)/3^n$ si $X(\omega) \in [k/3^n, (k+1)/3^n]$, $k = 0, \dots, 3^n - 1$. Les suites de variables aléatoires $(X_n^-)_{n \geq 0}$ et $(X_n^+)_{n \geq 0}$ convergent uniformément vers X ($\|X_n^\pm - X\|_\infty \leq 1/3^n$) et on constate qu'elles sont, pour $n \geq 0$ donné, uniformément réparties respectivement sur les extrémités gauches et droites des 2^n intervalles de C_n . Le calcul approché des espérances est alors immédiat :

```
vardef f expr x = x**2 enddef;
% remarquer que la normalisation n'apparaît qu'à la fin du calcul
def EX expr depth = EXapprox := EXsub(0,1,depth)/(2**depth); message
    "E[f(X)] = " & decimal(EXapprox); enddef;
vardef EXsub(expr u,v,depth) = if depth >0:
    EXsub(u,1/3[u,v],depth-1)+EXsub(2/3[u,v],v,depth-1)
    else: f(u) fi% ou f(v)
enddef;
EX 8;
```

⊞ L'espérance de $(X_n^-)^2$ s'écrit sous la forme d'une série :

$$\mathbb{E}[(X_n^-)^2] = \frac{1}{2^n} \sum_k \left(\frac{k}{3^n}\right)^2$$

ou si on sait que X peut s'écrire $\sum_{n \geq 1} \xi_n/3^n$ avec $(\xi_n)_{n \geq 1}$ variables aléatoires indépendantes uniformément distribuées sur $\{0, 2\}$, on a

$$\mathbb{E}[X^2] = \sum_{m,n \geq 1} \frac{\mathbb{E}[\xi_m \xi_n]}{3^{m+n}} = \sum_{m \geq 1} \frac{2}{(9)^m} + 2 \sum_{1 \leq m < n} \frac{1}{3^m 3^n} = \frac{2}{8} + \frac{2}{8} \times \frac{1}{2} = \frac{3}{8}.$$

En précisant tout ça proprement, ça devrait être jouable avec un peu d'effort. Le calcul numérique indique que $\mathbb{E}[X^2] = 3/8$ (comme prédit), donc $\text{Var}(X) = 3/8 - 1/4 = 1/8$. Notons que si U est de loi uniforme sur $[0, 1]$, $\mathbb{E}[U] = 1/2$, $\text{Var}(U) = 1/3 - 1/4 = 1/12$, ce qui indique bien que la mesure de Cantor est plus dispersée autour de sa valeur moyenne que ne l'est la loi uniforme.

1.2. PROCESSUS À VARIATION FINIE



TRAVAUX PRATIQUES N° 2. — INTÉGRALES STOCHASTIQUES

1. La méthode standard

Il n'y a pas beaucoup de choix pour calculer de manière approchée une intégrale stochastique $\int_0^T H_t dX_t$. On dispose d'une subdivision du temps $0 = t_1 < \dots < t_{n+1} = T$ (indexation cohérente avec SCILAB) fixée pour une raison ou une autre (nature ou simulation du processus X , dates de relevés réels, ...). Nous savons de plus que la non nullité éventuelle de la variation quadratique $[X, X]$ implique que les schémas numériques classiques (rectangles, trapèzes) donnent lieu à des intégrales stochastiques différentes.

Nous avons donc pour l'intégrale de Itô

$$(H \cdot X)_T = \int_0^T H_t dX_t \approx \sum_{i=1}^n H(t_i) \times (X(t_{i+1}) - X(t_i))$$

et pour l'intégrale de Stratonovich

$$(H \circ X)_T = \int_0^T H_t \circ dX_t \approx \sum_{i=1}^n \frac{1}{2} (H(t_i) + H(t_{i+1})) \times (X(t_{i+1}) - X(t_i))$$

Ce qu'on peut mettre en œuvre de manières diverses, par exemple

// La méthode standard

```
T = 1; n = 1000;
time = (T/n)*[0:n]';
dX = grand(n, 1, "nor", 0, sqrt(T/n)); // accroissements browniens standard
X = cumsum([0; dX], "r"); // brownien issu de 0. Noter que dX(i) = X(i+1)-X(i)
H = 2*X; // H_t=2X_{t-} (voir formules d'intégration par parties)
HdotX = zeros(n+1, 1);
HcircX = zeros(n+1, 1);
HdotX(1) = 0;
HcircX(1) = 0;
for i = 1:n;
    HdotX(i+1) = HdotX(i)+H(i)*(X(i+1)-X(i));
    HcircX(i+1) = HcircX(i)+(H(i)+H(i+1))/2*(X(i+1)-X(i));
end
```

Les boucles, bien que très déconseillées en SCILAB, assurent un bon contrôle des indices et la cohérence avec les définitions.


EXERCICE 1. — ☞ Saisir et exécuter ce morceau de programme.


☞ Vérifier que les contraintes suivantes sont satisfaites :

- Les « intégrales stochastiques » ont une valeur initiale nulle.
- Elles ont le même ensemble d'indices que le processus X .
- L'intégrale de Itô définit une martingale lorsque X est une martingale.

☞ Représenter sur un même graphique les trois processus X , $H \cdot X$ et $H \circ X$.

```
clf(); plot2d(time, [X, HdotX, HcircX]);
legend(["processus X"; "processus H.X"; "processus HoX"]);
halt("*** taper [entree] pour continuer ***\n");
```

EXERCICE 2. —  Définir une fonction SCILAB $\text{Ito}(\langle H \rangle, \langle X \rangle)$ retournant le processus $H \cdot X$. On s'imposera pour contrainte de ne pas employer de boucle et de faire dépendre la taille du vecteur (colonne) de sortie de celles des vecteurs (colonne) d'entrée en avertissant d'une incohérence éventuelle.


 Définir de même une fonction SCILAB $\text{Strato}(\langle H \rangle, \langle X \rangle)$ respectant des contraintes similaires.


2. Formules de changement de variables


Nous rappelons que si $X = (X_t)_{t \geq 0}$ est une semi-martingale réelle et $f : \mathbb{R} \rightarrow \mathbb{R}$ est une fonction de classe C^2 , alors la formule du changement de variable, ou formule de Itô, s'écrit


$$f(X_T) - f(X_0) = \int_0^T f'(X_{t-}) dX_t + \frac{1}{2} \int_0^T f''(X_{t-}) d[X, X]_t, \quad T \geq 0.$$


où $[X, X]$ est la variation quadratique de X qui est égale à sa projection duale prévisible $\langle X, X \rangle$ lorsque X est à trajectoires continues, et simplement $[X, X]_t = t$ lorsque X est un mouvement brownien linéaire. Aux difficultés techniques près, cette formule repose simplement sur la définition de l'intégrale stochastique comme limite de sommes de Riemann et sur la formule de Taylor. Elle doit pouvoir se retrouver sans trop de difficultés avec ces approximations. Et il devrait en être de même de la formule de changement de variables, qui est la formule ordinaire, dans le cadre de l'intégrale de Stratonovich.

EXERCICE 3. —  Reconnaître dans les exercices précédents les formules de changement de variables correspondantes. Comparer graphiquement les processus $f(X) - f(X_0)$, l'intégrale de Itô $f'(X) \cdot X$, le second membre de la formule de Itô et l'intégrale de Stratonovich $f'(X) \circ X$ pour la fonction f considérée.


 Pour $\langle G \rangle$ un vecteur colonne représentant un processus quelconque, à quoi correspondent les résultats de $\text{Ito}(\langle G \rangle, \text{time})$ et $\text{Strato}(\langle G \rangle, \text{time})$?

 Mettre à l'épreuve les approximations des intégrales stochastiques pour la fonction trigonométrique $x \mapsto \sin(2\pi x/T)$.

 Mettre à l'épreuve les approximations des intégrales stochastiques pour les fonctions exponentielles $x \mapsto e^x$ et $x \mapsto e^{-x}$.

 Qu'obtient-on pour $x \mapsto |x|$, $x \mapsto x^+ = \max(x, 0)$, et $x \mapsto -x^- = \min(x, 0)$? *Le temps local en 0 du mouvement brownien vient de faire son apparition...*

Remarque. — Le temps local du mouvement brownien en 0 est défini comme le processus croissant L^0 tel que $X_t^+ - X_0^+ = \int_0^t \mathbb{1}_{\{X_s > 0\}} dX_s + \frac{1}{2} L_t^0$. Il apparaît dans les formules de Itô–Tanaka liées à X^- , $|X|$, ... et définit une mesure positive non nulle portée par les zéros du brownien qui sont de mesure de Lebesgue nulle. L'homogénéité spatiale du mouvement brownien amène à considérer le temps local L^x en $x \in \mathbb{R}$ qui est un processus similaire à L^0 (le même quitte à changer la valeur initiale du processus).

 Dédire des formules d'approximation des intégrales stochastiques une approximation du temps local L^0 . Lorsque X est approché à l'aide de la marche aléatoire simple sur $\varepsilon\mathbb{Z}$, exprimer cette approximation en terme de *nombre de montées* ou de *nombre de descentes* du processus autour de $x = 0$.

TRAVAUX PRATIQUES N° 2. — INTÉGRALES STOCHASTIQUES, ÉLÉMENTS D'EXPLICATIONS

1. La méthode standard

⊞ Pour l'intégrale de Itô :

```
function Y = Ito(H, X)
    // local m n
    n = size(H, 1); m = size(X,1);
    if n+1 == m then
        mprintf("H et X de tailles compatibles ");
        mprintf("pour l'intégrale de Ito.\n");
    elseif n >= m then
        mprintf("des valeurs de H seront ignorees ");
        mprintf("pour l'intégrale de Ito.\n");
        n = m-1;
    else
        mprintf("des valeurs de X seront ignorees ");
        mprintf("pour l'intégrale de Ito.\n");
    end
    Y = [0; cumsum(H(1:n).*(X(2:n+1)-X(1:n)), "r")];
endfunction
```

⊞ Pour l'intégrale de Stratonovich :

```
function Y = Strato(H, X)
    // local m n
    n = size(H, 1); m = size(X,1);
    if n == m then
        mprintf("H et X de tailles compatibles ");
        mprintf("pour l'intégrale de Stratonovich.\n");
    elseif n > m then
        mprintf("des valeurs de H seront ignorees ");
        mprintf("pour l'intégrale de Stratonovich.\n");
        n = m;
    else
        mprintf("des valeurs de X seront ignorees ");
        mprintf("pour l'intégrale de Stratonovich.\n");
    end
    Y = [0; cumsum(0.5*(H(1:n-1)+H(2:n)).*(X(2:n)-X(1:n-1)), "r")];
endfunction
```

On vérifie

```
clf(); plot2d(time, [X, Ito(H, X), Strato(H, X)]);
legend(["processus X"; "processus Ito(H, X)"; "processus Strato(H, X)"]);
halt("*** taper [entree] pour continuer ***\n");
```

2. Formules de changement de variables

☞ Ici on a bien sûr $f(x) = x^2$.

```
// le processus
T = 1; n = 1000;
time = (T/n)*[0:n]';
dX = grand(n, 1, "nor", 0, sqrt(T/n));
X = cumsum([0; dX], "r");

// f(x) = x^2 (car f(x) = x est trop simple\dots)

fX = X.^2;
H = 2*X;
G = 2*ones(size(X));
I = Ito(H, X);
Z = time; // 0.5*Ito(G, time);
S = Strato(H, X);
clf(); plot2d(time, [fX-fX(1), I, I+Z, S]);
xlabel("X mouvement brownien, f(X) = X^2");
legend(["f(X)-f(Xo)"; "Int 2X dX"; "Int 2X dX + t"; "Strato 2X dX"]);
halt("*** taper [entree] pour continuer ***\n");
```

☞ $\text{Ito}(\langle G \rangle, \text{time})$ est une simple somme de Riemann (méthode des rectangles) approchant $\int_0^T G_t dt$, tandis que $\text{Strato}(\langle G \rangle, \text{time})$ correspond à l'approximation par la méthode des trapèzes.



```
// f(x) = sin(2pi*x/T)

fX = sin(2*pi*X/T);
H = 2*pi/T*cos(2*pi*X/T);
G = -(2*pi/T)^2*sin(2*pi*X/T);
I = Ito(H, X);
Z = 0.5*Strato(G, time); // mieux que Ito ?
S = Strato(H, X);
clf(); plot2d(time, [fX-fX(1), I, I+Z, S]);
xlabel("X mouvement brownien, f(X) = sin(2pi*X/T)");
legend(["f(X)-f(Xo)"; "Int f''(X) dX"; ..
"Int f''(X) dX + 1/2Int f''''(X) dt"; "Strato f''(X) dX"]);
halt("*** taper [entree] pour continuer ***\n");
```



```
// f(x) = exp(x)

fX = exp(X);
H = fX;
G = fX;
I = Ito(H, X);
Z = 0.5*Ito(G, time);
S = Strato(H, X);
clf(); plot2d(time, [fX-fX(1), I, I+Z, S]);
xlabel("X mouvement brownien, f(X) = exp(X)");
```



```

legend(["f(X)-f(Xo)"; "Int f''(X) dX"; ..
"Int f''(X) dX + 1/2Int f''''(X) dt"; "Strato f''(X) dX"]);
halt("*** taper [entree] pour continuer ***\n");

```



```

// f(x) = exp(-X)
fX = exp(-X);
H = -fX;
G = fX;
I = Ito(H, X);
Z = 0.5*Ito(G, time);
S = Strato(H, X);
clf(); plot2d(time, [fX-fX(1), I, I+Z, S]);
xtitle("X mouvement brownien, f(X) = exp(-X)");
legend(["f(X)-f(Xo)"; "Int f''(X) dX"; ..
"Int f''(X) dX + 1/2Int f''''(X) dt"; "Strato f''(X) dX"]);
halt("*** taper [entree] pour continuer ***\n");

```



```

// f(x) = exp(-X)
fX = abs(X);
for i = 1:n+1;
    if X(i) < 0 then H(i) = -1;
    elseif X(i) > 0 then H(i) = 1;
    else H(i) = 0; end
end
G = zeros(n+1, 1);
I = Ito(H, X);
Z = 0.5*Ito(G, time);
S = Strato(H, X);
clf(); plot2d(time, [fX-fX(1), I, I+Z, S]);
xtitle("X mouvement brownien, f(X) = |X|");
legend(["f(X)-f(Xo)"; "Int f''(X) dX"; ..
"Int f''(X) dX + 1/2Int f''''(X) dt"; "Strato f''(X) dX"]);
halt("*** taper [entree] pour continuer ***\n");

```

Attention! — L'intégrale de Itô est égale au second membre de la formule correspondante, donc les courbes (rouge et verte) correspondantes coïncident. Curieusement, l'intégrale de Stratonovich continue à donner des résultats convenables, elle ignore le temps local! Ceci peut se voir sur les schémas d'approximation (à préciser un jour).

```

// f(x) = x+
for i = 1:n+1;
    if X(i) < 0 then fX(i) = 0; H(i) = 0;
    else fX(i) = X(i); H(i) = 1; end
end
G = zeros(n+1, 1);
I = Ito(H, X);
Z = 0.5*Ito(G, time);
S = Strato(H, X);

```

```

clf(); plot2d(time, [fX-fX(1), I, I+Z, S]);
xlabel("X mouvement brownien, f(X) = X+");
legend(["f(X)-f(Xo)"; "Int f''(X) dX"; ..
"Int f''(X) dX + 1/2Int f''''(X) dt"; "Strato f''(X) dX"]);
halt("*** taper [entree] pour continuer ***\n");

```

Y a-t'il des différences lorsque la condition pour définir $\langle H \rangle$ est modifiée ? Non, nous sommes dans un cadre discret.

```

// f(x) = -x-
for i = 1:n+1;
    if X(i) > 0 then fX(i) = 0; H(i) = 0;
    else fX(i) = X(i); H(i) = -1; end
end
G = zeros(n+1, 1);
I = Ito(H, X);
Z = 0.5*Ito(G, time);
S = Strato(H, X);
clf(); plot2d(time, [fX-fX(1), I, I+Z, S]);
xlabel("X mouvement brownien, f(X) = -X-");
legend(["f(X)-f(Xo)"; "Int f''(X) dX"; ..
"Int f''(X) dX + 1/2Int f''''(X) dt"; "Strato f''(X) dX"]);
halt("*** taper [entree] pour continuer ***\n");

```

TRAVAUX PRATIQUES N° 3. — ÉQUATIONS DIFFÉRENTIELLES STOCHASTIQUES (1)

Les équations différentielles stochastiques sont en fait des équations intégrales

$$X_t = X_0 + \int_0^t H_s dB_s + \int_0^t K_s ds, \quad t \geq 0,$$

où $X = (X_t)_{t \geq 0}$ est le processus inconnu, $B = (B_t)_{t \geq 0}$ est une semi-martingale donnée et les processus prévisibles $H = (H_t)_{t \geq 0}$ et $K = (K_t)_{t \geq 0}$ sont des processus prévisibles dont l'expression peut dépendre directement de t , X_t , et $\omega \in \Omega$. Par commodité, on écrit sous forme différentielle

$$dX_t = H_t dB_t + K_t dt, \quad t \geq 0.$$

Cette écriture suggère une méthode de résolution approchée simple identique au cas des équations différentielles ordinaires : *la méthode d'Euler*.

Nous nous plaçons sur l'intervalle de temps $[0, T]$ dont nous considérons une subdivision régulière $0 = t_0 < t_1 < \dots < t_{n-1} < t_n = T$ avec $t_i = T \times i/n$. Notons $\Delta B_i = B_{t_i} - B_{t_{i-1}}$, $\Delta X_i = X_{t_i} - X_{t_{i-1}}$, $H_i = H_{t_i}$ et $K_i = K_{t_i}$, et $\Delta t = T/n$. En gardant en tête qu'il s'agit du calcul intégral de Itô, l'équation discrétisée s'écrit

$$\Delta X_i = H_{i-1} \Delta B_i + K_{i-1} \Delta t, \quad i = 1, \dots, n,$$

ou encore

$$X_i = X_{i-1} + H_{i-1} \Delta B_i + K_{i-1} \Delta t, \quad i = 1, \dots, n,$$

avec $X_i = X_{t_i}$.

Remarque. — Voulant faire coïncider l'indice i avec l'instant t_i , l'indexation des vecteurs SCILAB sera décalé d'une unité, par exemple : $X_i = \mathbf{X}(i+1)$, $i = 0, \dots, n$; $H_{i-1} = \mathbf{H}(i)$, $i = 1, \dots, n$; $\Delta B_i = \mathbf{dB}(i)$, $i = 1, \dots, n$.

EXERCICE 1. — Soient $B = (B_t)_{t \geq 0}$ le mouvement brownien standard unidimensionnel issu de 0 et $a \in \mathbb{R}$ une constante. Nous considérons l'équation différentielle stochastique autonome $(H_t = f(X_{t-}), K_t = g(X_{t-}))$, linéaire

$$dX_t = aX_{t-} dB_t, \quad t \geq 0. \tag{E_1}$$

On peut lire que sa solution (exponentielle stochastique du mouvement brownien) en un sens fort est unique et qu'elle est donnée par

$$X_t = X_0 \times \exp(aB_t - a^2/2 \times t), \quad t \geq 0. \tag{S_1}$$

Elle est continue et le « $t-$ » aurait pu être remplacé par « t ». Nous ne le faisons pas car cette précision est importante pour la discrétisation.

☞ À l'aide de la formule de Itô, vérifier que (S₁) fournit une solution de (E₁). Utiliser un théorème d'unicité pour conclure.

☞ Nous allons comparer la solution explicite avec la solution approchée obtenue par la méthode d'Euler.

`clear; mode(0);`

```

T = 1; n = 100; a = 1; // par exemple
t = [0:n]'*T/n; dt = T/n;
dB = grand(n, 1, "nor", 0, sqrt(dt)); // accroissements N(0, dt)
B = cumsum([0; dB]); // mouvement brownien issu de 0
Xzero = 1;
Xex = Xzero*exp(a*B-a^2/2*t); // calcul vectorialisé
Xapp = zeros(n+1, 1); Xapp(1) = Xzero;
for i = 1:n; Xapp(i+1) = Xapp(i)*(1+a*dB(i)); end
clf();
plot2d(t, [Xex, Xapp]);
legend(["solution exacte", "solution approchée"], 4);
// le '4' indique d'écrire la légende en bas à droite
xlabel("dX = aX dB");

```

Effectuer plusieurs simulations (modifier n en particulier) et commenter.

☉ Calculer explicitement la solution approchée. Justifier grossièrement pourquoi celle-ci converge bien vers l'exponentielle stochastique.

☉ Qu'en aurait-il été des solutions exactes ou approchées de (E_1) si les trajectoires de B avait été régulières (C^1 par exemple) ?

☉ Vous allons étudier la vitesse de convergence de la méthode de discrétisation. Les paramètres de simulation T , a , X_0 seront inchangés. Pour

$$n = 10, 20, \dots, 50, 100, \dots, 500, 600, \dots, 1000,$$

nous utiliserons $N = 1000$ trajectoires afin d'estimer

$$\sigma(n) = \sqrt{\mathbb{E}[(X_T - X_{\text{app},T})^2]}$$

l'erreur quadratique entre la solution exacte et la solution approchée au temps T . Compte tenu de notre propos, il ne sera pas la peine de conserver toutes les informations sur les trajectoires, seulement ce qui est nécessaire pour obtenir les valeurs terminales et estimer l'erreur quadratique par l'écart-type observé (avec le dénominateur $N - 1$ pour fixer les idées). La qualité de la programmation sera appréciée.

Tracer la suite des estimations des erreurs quadratiques en fonction de n . L'affirmation selon laquelle elle se comporte en $1/\sqrt{n}$ vous paraît-elle fondée ? On pourrait passer au carré et effectuer une régression linéaire :

$$y = ax + b, \quad a = \frac{\text{cov}(x, y)}{s_x^2}, \quad b = \bar{y} - a \times \bar{x}, \quad r = \frac{\text{cov}(x, y)}{s_x \times s_y}$$

(variances et écart-types non corrigés).

EXERCICE 2. — Reprendre l'exercice précédent en remplaçant les accroissements gaussiens par les accroissements d'une marche aléatoire simple symétrique sur \mathbb{Z} convenablement normalisée (principe d'invariance de Donsker). Cette méthode est parfois qualifiée de méthode d'Euler faible.

TRAVAUX PRATIQUES N° 3. — ÉQUATIONS DIFFÉRENTIELLES STOCHASTIQUES (1), ÉLÉMENTS D'EXPLICATIONS

EXERCICE 1. — \otimes Le mouvement brownien étant supposé issu de 0, la condition initiale est satisfaite. On a $X_t = f(B_t, t)$ et ainsi

$$\begin{aligned} X_t - X_0 &= \int_0^t \frac{\partial}{\partial x} f(B_{s-}, s) dB_s + \int_0^t \frac{\partial}{\partial t} f(B_{s-}, s) ds + \int_0^t \frac{1}{2} \frac{\partial^2}{\partial x^2} f(B_{s-}, s) ds \\ &= \int_0^t a X_{s-} dB_s - \int_0^t a^2/2 \times X_{s-} ds + \int_0^t a^2/2 \times X_{s-} ds \\ &= \int_0^t a X_{s-} dB_s, \end{aligned}$$

donc (E_1) est vérifiée. Maintenant, cette équation s'écrit $dX_t = f(X_{t-}) dB_t$ avec f globalement lipschitzienne (de constante de Lischitz a), donc la solution (existe et) est unique à processus évanescents près. Nous avons donc la solution.

\otimes Rien à dire.

\otimes On a

$$X_n = X_{n-1}(1 + a \Delta B_n) = \dots = X_0 \times \prod_{i=1}^n (1 + a \Delta B_i).$$

Si les ΔB_i sont petits (...)

$$\begin{aligned} X_n &= X_0 \times \exp\left(\sum_{i=1}^n \ln(1 + a \Delta B_i)\right) = X_0 \times \exp\left(\sum_{i=1}^n a \Delta B_i - \frac{1}{2} a^2 \Delta B_i^2 + O(\Delta B_i^3)\right) \\ &= X_0 \times \exp\left(a \sum_{i=1}^n \Delta B_i - \frac{1}{2} a^2 \sum_{i=1}^n \Delta B_i^2 + \sum_{i=1}^n O(\Delta B_i^3)\right) \\ &= X_0 \times \exp\left(a B_T - \frac{1}{2} a^2 T + O(1)\right) \longrightarrow X_T \quad \text{quand } n \rightarrow \infty, \end{aligned}$$

puisque les sommes de $O(\Delta B_i^3)$ sont dominées par les variations d'ordre 3 qui tendent vers 0. Nous retrouvons bien la solution. On a l'impression qu'avec un peu de travail, nous pourrions presque explicitement mesurer l'erreur de discrétisation. Il se trouve qu'elle est en $1/\sqrt{n}$ (erreur quadratique).

\otimes Si les trajectoires sont régulières, on a $dB_t = B'_t dt$ et l'équation s'écrit $dX_t = (aB'_t)X_t dt$ ou encore

$$X'_t = (aB'_t)X_t$$

dont la solution (équation différentielle ordinaire à ω fixé, globalement lipschitzienne sur les intervalles de temps bornés) est

$$X_t = X_0 \times \exp(aB_t).$$

La discrétisation est identique, mais lorsque n tend vers l'infini, les variations d'ordre $p > 1$ de B_t tendent vers 0, en particulier pour $p = 2$. On retrouve alors à la limite la solution plus haut.

TRAVAUX PRATIQUES N° 3. — ÉQUATIONS DIFFÉRENTIELLES STOCHASTIQUES (2)

1. Méthodes numériques pour les équations différentielles ordinaires

Exemple. — L'exponentielle...

2. Méthodes numériques pour les équations différentielles stochastiques

Exemple. — L'exponentielle stochastique...

3. L'équation logistique

Dans le secondaire, les équations différentielles ordinaires du premier ordre sont souvent illustrées par le modèle de croissance de population exponentielle : si r désigne l'intensité ou le taux de croissance de la population, $x(t)$ sa taille au temps t (considérée comme un réel!), on décrète que

$$\frac{dx}{dt}(t) = rx(t), \quad x(t_0) = x_0,$$

pour obtenir comme solution $x(t) = x_0 \exp(r(t-t_0))$. Le défaut de cette modélisation est que la population croît très vite et tendra vers l'infini.

Remarque. — Le modèle à temps discret correspondant est $x_{n+1} = (1+r) \times x_n$, $n \geq 0$, qui donne $x_n = x_0(1+r)^n$ et est donc aussi à croissance exponentielle. Le passage du discret au continu se fait en remplaçant r par r/N , et en posant $t = n/N$ ou $n = t \times N$: on subdivise le temps entier en N périodes. En faisant tendre N vers l'infini à t fixé, on retrouve la solution du modèle à temps continu.

Un modèle de croissance de population plus fin est donné par le modèle de Verhulst–Malthus¹ :

$$\frac{dx}{dt}(t) = rx(t) \times (K - x(t))$$

dont les solutions sont de la forme


$$x(t) = \frac{\exp(rt/K - c)}{1 + \exp(rt/K - c)}.$$

Lorsque la population est de taille $x(t)$ faible, c'est-à-dire petite devant la limite K , on a quasiment une exponentielle. Mais après elle se retrouve limitée par les ressources et prend

1. Pierre-François Verhulst (28 octobre 1804 – 15 février 1849, Bruxelles) est un mathématicien belge.

Il étudia les mathématiques sous la direction de Quetelet. Inspiré par l'« Essai sur le principe de population » de Thomas Malthus, il proposa en 1838 le modèle de Verhulst, décrivant l'évolution des populations animales grâce à un modèle qui ne soit pas exponentiel. C'est dans la publication de 1845 qu'il nomme cette courbe « logistique » sans donner l'explication de ce terme.

l'allure d'une « sigmoïde » ou courbe logistique d'asymptote en $+\infty$ d'équation $y = K$.



$$dX_t = rX_t(K - X_t) dt + rX_t dB_t,$$